
ДОДАТОК А 

Апробація результатів роботи 

ДОДАТОК Б 

Програмний код 

 

 

//Розпіновка 

const int tempPin = A0; //Датчик температури 

const int irSensorPin = 3; //Інфрачервоний датчик перешкод 

const int ledPin = 4; //Світлодіод 

const int motorPin1 = 5; //Мотор-редуктор (+) 

const int motorPin2 = 6; //Мотор-редуктор (-) 

const int pumpPin1 = 7; //Насос (+) 

const int pumpPin2 = 8; //Насос (-) 

//Оголошення змінних 

bool ledState = HIGH; //Стан світлодіода 

bool systemRunning = true; //Стан системи 

unsigned long lastTempCheck = 0; //Час останньої перевірки температури 

unsigned long irDetectedTime = 0; //Час виявлення перешкоди 

bool irHandled = false; //Чи оброблено виявлення перешкоди 

bool pumpRunning = false; //Чи працює насос 

unsigned long pumpStartTime = 0; //Час запуску насосу 

 

void setup() { 

  Serial.begin(9600); //Ініціалізація послідовного порту 

  pinMode(ledPin, OUTPUT); 

  pinMode(irSensorPin, INPUT); //Інфрачервоний датчик 

  pinMode(motorPin1, OUTPUT); 

  pinMode(motorPin2, OUTPUT); 

  pinMode(pumpPin1, OUTPUT); 

  pinMode(pumpPin2, OUTPUT); 

  digitalWrite(ledPin, HIGH); //Світлодіод увімкнено 

  digitalWrite(motorPin1, HIGH); //Мотор-редуктор увімкнено 

  digitalWrite(motorPin2, LOW); 

  digitalWrite(pumpPin1, LOW); //Насос вимкнено 

  digitalWrite(pumpPin2, LOW); 

} 

 

void loop() { 

  //Перевірка температури кожні 500 мс 

  if (millis() - lastTempCheck >= 500) { 

    float temperature = readTemperature(); 

    lastTempCheck = millis(); 

    //Вивід значень температури до послідовного порту 

    Serial.print("Temperature: "); 

    Serial.print(temperature); 

    Serial.println("°C"); 

    //Перевірка значення температури 

    if (temperature < 0 || temperature > 40) { 

      systemRunning = false; //Зупинка системи 

      stopAll(); //Зупинка моторів 

      blinkLED(); //Блимання світлодіода 

      return; //Вихід з циклу 

    } else { 

      systemRunning = true; //Система працює 

      digitalWrite(ledPin, HIGH); //Світлодіод увімкнено 

    } 

  } 

 

  //Робота системи 

  if (systemRunning) { 

    //Перевірка інфрачервоного датчика 

    if (digitalRead(irSensorPin) == LOW && !irHandled) { //Перешкоду виявлено 

      if (irDetectedTime == 0) { 

        irDetectedTime = millis(); //Початок відліку часу 

        digitalWrite(motorPin1, LOW); //Зупинка мотор-редуктору 

        digitalWrite(motorPin2, LOW); 

        digitalWrite(pumpPin1, HIGH); //Увімкнення насосу 

        digitalWrite(pumpPin2, LOW); 

        pumpRunning = true; 

        pumpStartTime = millis(); 

      } 

    } 

 

    //Перевка часу роботи насосу 

    if (pumpRunning && millis() - pumpStartTime >= 2000) { 

      digitalWrite(pumpPin1, LOW); //Вимикаємо насос 

      digitalWrite(pumpPin2, LOW); 

      pumpRunning = false; 

      irHandled = true; //Виявлення оброблено 

      //Відновлення роботи мотор-редуктору одразу після зупинки насоса 

      digitalWrite(motorPin1, HIGH); 

      digitalWrite(motorPin2, LOW); 

    } 

 

    //Якщо перешкода все ще виявлена після 2 секунд роботи насосу + 1 секунду 

    if (irHandled && digitalRead(irSensorPin) == LOW) { 

      if (millis() - irDetectedTime >= 3000) { //3 секунди від виявлення 

        systemRunning = false; //Зупинка системи 

        stopAll(); //Зупинка моторів 

        blinkLED(); //Блимання світлодіоду 

      } 

    } else if (irHandled && digitalRead(irSensorPin) == HIGH) { 

      //Скидання стану, якщо перешкода зникла 

      irHandled = false; 

      irDetectedTime = 0; 

    } 

  } 

} 

//Зчитування температури 

float readTemperature() { 

  int sensorValue = analogRead(tempPin); 

  float voltage = sensorValue * (5.0 / 1023.0); //Конвертація в вольти 

  float temperature = (voltage - 0.5) * 100; //Конвертація в градуси Цельсія 

  return temperature; 

} 

//Зупинка всього 

void stopAll() { 

  digitalWrite(motorPin1, LOW); //Зупинка мотор-редуктора 

  digitalWrite(motorPin2, LOW); 

  digitalWrite(pumpPin1, LOW); //Зупинка насоса 

  digitalWrite(pumpPin2, LOW); 

} 

//Блимання світлодіоду 

void blinkLED() { 

  unsigned long currentTime = millis(); 

  if (currentTime % 1000 < 500) { 

    digitalWrite(ledPin, HIGH); 

  } else { 

    digitalWrite(ledPin, LOW); 

  } 

} 

ДОДАТОК В 

Демонстраційний матеріал 

доку-

Позначення Найменування 
Додаткові 

відомості 

  Текстові документи  

    

1 ГЮИК. 306588.039 ПЗ Пояснювальна записка А4, 82 с. 

    

  Додаткові матеріали  

    

2  Програмний код для  Arduino А4, 2 с. 

